# $Id: PKGBUILD 41765 2011-03-08 20:18:32Z spupykin $
# Maintainer: Florian Pritz <bluewind@xinu.at>
# Contributor: Sergej Pupykin <pupykin.s+arch@gmail.com>
# Contributor: Francois Charette <firmicus@gmx.net>

pkgname=perl-mail-dkim
pkgver=0.40
pkgrel=1
pkgdesc="Implements DomainKeys Identified Mail (DKIM)"
arch=('any')
url="http://search.cpan.org/dist/Mail-DKIM"
license=('GPL' 'PerlArtistic')
depends=('perl-crypt-openssl-rsa>=0.24' 'perl-digest-sha1' 'perl-mailtools' 'perl-net-dns')
options=('!emptydirs')
source=(http://www.cpan.org/authors/id/J/JA/JASLONG/Mail-DKIM-$pkgver.tar.gz)
md5sums=('9ceb8a0397aff83928e7a9760b1b190f')

build() {
  cd "$srcdir/Mail-DKIM-$pkgver"
  PERL_MM_USE_DEFAULT=1 perl Makefile.PL INSTALLDIRS=vendor
  make
}

check() {
  cd "$srcdir/Mail-DKIM-$pkgver"

  make test
}

package() {
  cd "$srcdir/Mail-DKIM-$pkgver"

  make install DESTDIR="$pkgdir"
  find "$pkgdir" -name '.packlist' -delete
  find "$pkgdir" -name '*.pod' -delete
}
